/*
Maria Abascal	
Proposal title: Black–White Relations in the Wake of Hispanic Growth: Generosity and Identification


HYPOTHESES

Stated-Hyp1: "How will Hispanic growth affect relations between US Blacks and Whites?"... "Whites may generalize the threat posed by Hispanics to all non-Whites and attempt to hoard resources from all out-group members; this would explain both their smaller contributions to Blacks and greater contributions to Whites in the treatment––as opposed to the control––condition."

	Test-Hyp1: Among whites, contributions to blacks will be smaller in the Hispanic treatment condition vs. control

	Test-Hyp2: Among whites, contributions to whites will be larger in the Hispanic treatment condition vs. control
	
Stated-Hyp2: "How will Hispanic growth affect relations between US Blacks and Whites?"..."For Blacks, the Hispanic growth prime may heighten perceived zero-sum competition with Hispanics relative to Whites; this would explain Blacks' relatively larger contributions to Whites in the treatment condition" 

	Test-Hyp3: Among blacks, contributions to blacks will vary in the Hispanic treatment condition vs. control

	Test-Hyp4: Among blacks, contributions to whites will vary in the Hispanic treatment condition vs. control

Stated-Hyp3: "I will evaluate the mediating role of racial and national stereotypes of Hispanics."... "Whites in the treatment condition should report that 1) a smaller share of Hispanics are racially white, and 2) a smaller share of all Americans are white; these perceptions should predict stronger racial identification and less equitable contributions."
	
	Test-Hyp5: Among whites, perception of hispanics as whites will be lower in treatment vs. control condition
	
	Test-Hyp6: Among whites, perception of the share of whites in the US will be smaller in treatment vs. control condition
	
	Test-Hyp7: Among whites, lower perception of hispanics as whites will predict stronger racial identification

	Test-Hyp8: Among whites, lower perception of share of whites in the US will predict lower contributions (regardless of recipient race)

	
Stated-Hyp4: "I predict 1) the impact of the hispanic growth prime will be relatively smaller among subjects in counties where hispanics make up a large or growing share of the population, and 2) subjects in counties where hispanics are predominantly foreign-born or of Mexican or Central-American origin will likely report that a larger share of Hispanics are non-white or foreign born in response to the Hispanic growth prime"

********************************************************************************
NOTES
- Stated hypothesis 4 cannot be tested because we don't have the racial composition of counties. We do have the fips codes, but it will take >15 minutes to search for, clean, and merge appropriate county-level data.
*/

clear all
use "Abascal635.dta", clear

********************************************************************************

* INDICATORS OF EXPERIMENTAL MANIPULATIONS

* treated group
	tab XTESS187
	recode XTESS187 (1/2=1) (3/4=0), gen(treated)
	tab treated

* race of the reciever
	recode XTESS187 (1 3=1 "reciever white") (2 4=0 "reciever black"), gen(rec_white)
	tab rec_white
	
* OUTCOME MEASURES

* money given to reciever
	tab Q4
	recode Q4 (-1=.r), gen(amount)
	tab amount

 	
* MODERATORS

* respondent race
	tab DOV_Respondent_Race
	recode DOV_Respondent_Race (1=1) (2=0), gen(resp_white)
	tab resp_white
	
* perception of share of whites in the us
	tab Q6_Whites
	recode Q6_Whites (-1=.r), gen(share_whites)

* perception of hispanics as whites	
	* what percentage of all Hispanics/Latinos in the United States belong to each group? - WHITES
	tab Q7_Whites
	recode Q7_Whites (-1=.r), gen(hispanics_whites)
	
* racial identification	
	/*
	Q11. Do you agree or disagree with the following statement? Being [IF DOV_RESP_RACE=1 INSERT "White".  IF DOV_RESP_RACE=2 INSERT "Black/African-American"] is important to the way I think of myself as a person.
	*/
	tab Q11
	recode Q11 (-1=.) (5=1) (4=2) (3=3) (2=4) (1=5), gen(racialid)
	tab racialid
	
********************************************************************************

* ANALYSIS
	

*	Test-Hyp1: Among whites, contributions to blacks will be smaller in the Hispanic treatment condition vs. control
	reg amount i.treated if resp_white==1& rec_white==0
	// reject. 0.862
	tess 1.treated -, init(Abascal635)
	
*	Test-Hyp2: Among whites, contributions to whites will be larger in the Hispanic treatment condition vs. control
	reg amount i.treated if resp_white==1& rec_white==1
	// reject. 0.911	
	tess 1.treated +

*	Test-Hyp3: Among blacks, contributions to blacks will vary in the Hispanic treatment condition vs. control
	reg amount i.treated if resp_white==0& rec_white==0
	// reject. 0.434
	tess 1.treated
	
*	Test-Hyp4: Among blacks, contributions to whites will vary in the Hispanic treatment condition vs. control
	reg amount i.treated if resp_white==0& rec_white==1
	// reject. 0.964	
	tess 1.treated 
	
* 	Test-Hyp5: Among whites, perception of hispanics as whites will be lower in treatment vs. control condition
	reg hispanics_whites i.treated if resp_white==1
	// reject. 0.474
	tess 1.treated -, bonf(4)	
	
*	Test-Hyp6: Among whites, perception of the share of whites in the US will be smaller in treatment vs. control condition
	reg share_whites i.treated if resp_white==1
	// reject. 0.071	
	tess 1.treated -, bonf(4)	
	
*	Test-Hyp7: Among whites, lower perception of hispanics as whites will predict stronger racial identification
	reg racialid hispanics_whites if resp_white==1
	// reject. 0.928
	tess hispanics_whites -, bonf(4)	
	
*	Test-Hyp8: Among whites, lower perception of share of whites in the US will predict lower contributions (regardless of recipient race)
	reg amount share_whites if resp_white==1
	// reject. 0.135
	tess share_whites +, bonf(4)	
	
